Using Maple to check partial derivatives

© Mike May, S.J., 2006 - maymk@slu.edu

Edited by Russell Blyth - blythrd@slu.edu

We can use Maple to compute derivatives for us, thus letting us check our hand computations. To show how to do this we first define a function.

```
> f := (x,y) -> x^2+3*x*y+5*x^3+2*sin(exp((y+1)/y^2));
```

$$f := (x, y) \to x^2 + 3\,x\,y + 5\,x^3 + 2\,\sin\left(e^{\frac{y+1}{y^2}}\right) \tag{1}$$

We can then find a partial derivative with the diff command.

```
> fx := diff(f(x,y),x);
```

$$fx := 2\,x + 3\,y + 15\,x^2 \tag{2}$$

We can take second derivatives by either taking the partial of the partial, or by using diff with two x's.

```
> fxx := diff(f(x,y),x,x);
  fxxA := diff(fx,x);
```

$$fxx := 2 + 30\,x$$
$$fxxA := 2 + 30\,x \tag{3}$$

Third partials work much the same way

```
> fxxx := diff(f(x,y),x,x,x);
  fxxxA := diff(fxx,x);
  fxxAx := diff(fxxA,x);
```

$$fxxx := 30$$
$$fxxxA := 30$$
$$fxxAx := 30 \tag{4}$$

With our function we can also take the partial derivative with respect to y.

```
> fy := diff(f(x,y),y);
```

$$fy := 3\,x + 2\,\cos\left(e^{\frac{y+1}{y^2}}\right)\left(\frac{1}{y^2} - \frac{2\,(y+1)}{y^3}\right)e^{\frac{y+1}{y^2}} \tag{5}$$

This is messy enough that we would like to see a worked solution rather than just the answer. For that we load the Student[Calculus1] package and use the DiffTutor command.

```
> with(Student[Calculus1]);
```

$$[AntiderivativePlot, AntiderivativeTutor, ApproximateInt, ApproximateIntTutor, ArcLength, ArcLengthTutor, Asymptotes, Clear, CriticalPoints, CurveAnalysisTutor, DerivativePlot, DerivativeTutor, DiffTutor, ExtremePoints, FunctionAverage, FunctionAverageTutor, FunctionChart, FunctionPlot, GetMessage, GetNumProblems, GetProblem, Hint, InflectionPoints, IntTutor, Integrand, InversePlot, InverseTutor, LimitTutor, MeanValueTheorem, MeanValueTheoremTutor, NewtonQuotient, NewtonsMethod, NewtonsMethodTutor, PointInterpolation, RiemannSum, RollesTheorem, Roots, Rule, \tag{6}$$

Show, ShowIncomplete, ShowSteps, Summand, SurfaceOfRevolution, SurfaceOfRevolutionTutor, Tangent, TangentSecantTutor, TangentTutor, TaylorApproximation, TaylorApproximationTutor, Understand, Undo, VolumeOfRevolution, VolumeOfRevolutionTutor, WhatProblem]

```
> TutorAnswer := DiffTutor(f(x,y),y);
```

$$TutorAnswer := \frac{\partial}{\partial y}\left(x^2+3xy+5x^3+2\sin\left(e^{\frac{y+1}{y^2}}\right)\right) = 3x + \frac{2\cos\left(e^{\frac{y+1}{y^2}}\right)e^{\frac{y+1}{y^2}}\left(y^2-2(y+1)y\right)}{y^4} \tag{7}$$

```
> TutorAnswer;
```

$$\frac{\partial}{\partial y}\left(x^2+3xy+5x^3+2\sin\left(e^{\frac{y+1}{y^2}}\right)\right) = 3x+\frac{2\cos\left(e^{\frac{y+1}{y^2}}\right)e^{\frac{y+1}{y^2}}\left(y^2-2(y+1)y\right)}{y^4} \tag{8}$$

To work with the right hand side of TutorAnswer we use the rhs command.

```
> Fy := rhs(TutorAnswer);
```

$$Fy := 3x+\frac{2\cos\left(e^{\frac{y+1}{y^2}}\right)e^{\frac{y+1}{y^2}}\left(y^2-2(y+1)y\right)}{y^4} \tag{9}$$

This lets us find mixed partials.

```
> Fyx := diff(Fy,x);
  fyx := diff(f(x,y),y,x);
```

$$Fyx := 3$$
$$fyx := 3 \tag{10}$$

```
>
```